Pan American Advanced Studies Institute Program:
Process Modeling and Optimization for Energy and Sustainability

GAMS Workshop #2

Example 2

Consider a problem similar to the one addressed earlier, but with 11 chemicals and 6 reactions:

Products P1, P2, P3, P4, P5 and P6 can be produced from raw materials A, B, C, D, and E via reactions R1, R2, … R6 as follows:

$$2A + 3B \rightarrow P1 + P2 \qquad \text{(R1)}$$

$$1A + 4B \rightarrow P3 \qquad \text{(R2)}$$

$$2C + 3D \rightarrow P4 \qquad \text{(R3)}$$

$$2C + 2P1 \rightarrow P3 + P5 \qquad \text{(R4)}$$

$$6E \rightarrow P6 \qquad \text{(R5)}$$

$$2D + 3P2 \rightarrow 3P6 \qquad \text{(R6)}$$

Given the availability (a_i) of raw materials, the minimum demands (d_i) of final products and the prices (p_i) of chemicals in the following table, what is the optimal production mix?.

i	A	B	C	D	E	P1	P2	P3	P4	P5	P6
a_i	20	30	25	18	24	-	-	-	-	-	-
d_i	-	-	-	-	-	2	3	2	3	4	5
p_i	3	4	5	6	5	35	45	30	50	40	45

The problem can be represented as a network (similar to the one for the previous example), where we consider the extend n_j of reaction j, the molar flow F_{ij} of chemical i in reaction j, and the total flow F_i of chemical i.

The mathematical formulation of our problem can be expressed via constraints (C1) – (C4) and (OBJ), where a subscript indicates an index for a ***set***, and $RM \subseteq I$ is the subset of raw materials and $FP \subseteq I$ is the subset of final products.

Mass balance constraints can be expressed via the extend of reaction and stoichiometric coefficients c_{ij}:

$$F_{ij} = c_{ij}\, n_j \quad \forall i,j \qquad \text{(C1)}$$

and the total flows can be expressed as the sum of individual flows:

$$F_i = \Sigma_j\, F_{ij} \quad \forall j \qquad \text{(C2)}$$

We also have constraints on the availability of raw materials:

$$- F_i \leq a_i \quad \forall i \in RM \qquad \text{(C3)}$$

and demand satisfaction constraints:

$$F_i \geq d_i \quad \forall i \in FP \qquad \text{(C4)}$$

The objective function is the maximization of profit:

$$max\;\; z = \Sigma_i\, c_i F_i \qquad \text{(OBJ)}$$

Note that variables F_{ij} and F_i are unrestricted, while variables n_j are nonnegative.
To develop a more concise formulation in GAMS, we can use keywords SETS and PARAMETERS. SETS are used to define families (sets) of objects and parameters are used to declare data in a general manner. In this specific example, we can define the set of chemicals $I = \{A, B, C, D, E, P1, P2, P3, P4, P5\}$ and the set of reactions J = {R1, R2, R3, R4, R5, R6}; the declaration of these sets is carried out in lines 6-7 of the GAMS file (see next page). We can also define the following arrays of data (lines 11-23):

a(i) = a_i - availability of chemical $i \in I$
d(i) = d_i - minimum demand for chemical $i \in I$
p(i) = p_i - price of chemical $i \in I$
c(i,j) = c_{ij} – stoichiometric coefficient of chemical I in reaction j

Furthermore, the decision variables are defined as arrays (see lines 25-30) using the keyword VARIABLES, and constraints C1 – C4 are expressed over (sub)sets (see lines 32-38).

The corresponding GAMS file is given in the table in the next page, where:

- We declare the sets of chemicals and reactions using the keyword SETS. The index i refers to the elements of the set of chemicals, and j refers to the elements to the set of reactions; a brief description of the set follows (it can be omitted); the elements of the set are given as a list.
- Next we define the subset of raw materials (irm) and the subset of final products (ifp); in parentheses we should identify the superset in which the elements of the subset belong.
- The declaration of data is performed using the keywords PARAMETERS and TABLE. Three arrays a(i), d(i), and p(i) are declared; a short description follows; the i^{th} element of an array refers to the i^{th} element of the corresponding set; the values of the elements are given in a list. Two-dimensional arrays can be declared as tables.
- Variables are defined as 1-D [Ft(i), n(j)] and 2-D [F(I,j)] arrays with a short description in lines 26-28; variables n(j) as defined as positive (nonnegative in reality) in line 30.
- In line 32, we declare the four constraints and the objective function of our optimization problem, and in lines 34-38 we enter these constraints using the sets, parameters and variables defined earlier.
- Note that constraint C1 is expressed for every (i,j) pair, constraint C2 is expressed for every chemical, and constraints C3 and C4 are expressed for the elements in subsets irm and ifp, respectively. In OBJ we consider the sum of flows, i.e. the amounts of raw materials we have to purchase (negative flows) and the amounts we sell (positive flows).

Table 2: “Advanced” GAMS model for Example 2.

```
$ TITLE Example 2
$ OFFSYMXREF
$ OFFSYMLIST

SETS
i       Chemicals       /A,B,C,D,E,P1,P2,P3,P4,P5,P6/
j       Reactions       /R1,R2,R3,R4,R5,R6/
irm(i)  Raw materials   /A,B,C,D,E/
ifp(i)  Final products  /P1,P2,P3,P4,P5,P6/;

PARAMETERS
a(i)    Availability    /A 20, B 30, C 25, D 18, E 24 /
d(i)    Demand          /P1 2, P2 3, P3 2, P4 3, P5 4, P6 5/
p(i)    Price           /A 3, B 4, C 5, D 6, E 5, … /;

TABLE c(j,i)
    A  B  C  D  E  P1 P2 P3 P4 P5 P6
R1  -2 -3          1  1
R2  -1 -4                1
R3        -2 -3            1
R4        -2             1     1
R5              -6                1
R6           -2       -3          3;

VARIABLES
F(i,j)  Flow of chemical i in reaction j
Ft(i)   Total flow of chemical i
n(j)    Extend of reaction j
Prof    Profit;
POSITIVE VARIABLE n;

EQUATIONS C1, C2, C3, C4, OBJ;

C1(i,j)..       F(i,j) =E= c(j,i)*n(j);
C2(i)..         Ft(i) =E= sum(j, F(i,j) );
C3(i)$irm(i)..  -Ft(i) =L= a(i);
C4(i)$ifp(i)..  Ft(i) =G= d(i);
OBJ..           Prof =E= sum(i, p(i)*Ft(i) );

MODEL P1 /ALL/;

OPTION LIMROW = 0;
OPTION LIMCOL = 0;
OPTION SOLPRINT = off;

SOLVE P1 USING LP MAXIMIZING Prof;
DISPLAY n.L, F.L, Ft.L;
```

Example 3

Consider the problem of assigning process streams (A, B, C and D) to heat exchangers (1, 2, 3, and 4) (from pp. 409-410 of "Optimization of Chemical Process" (1st ed.) by Edgar and Himmelblau). The cost of assigning a stream to an exchanger is given below (in $\$10^3$). What is the optimal solution if:

a) all assignments are allowed
b) assignments with a cost smaller than or equal to $10,000 are forbidden.

Streams	Exchangers			
	1	2	3	4
A	94	1	54	68
B	74	10	88	82
C	73	88	8	76
D	11	74	81	21

The GAMS file Example3.gms has two models P3A and P3B for cases (a) and (b), respectively. Note the use of the "dollar operation" in constraints C3 and C5 to express the condition that some Y's should not be taken into account (you can find more in The User's Manual).

Table 3: GAMS models for Example 3.

```
$ TITLE Example3
$ OFFSYMXREF
$ OFFSYMLIST

SET
i       Streams         /A, B, C, D/
j       Exchangers      /1, 2, 3, 4/;

TABLE C(i,j) Cost
      1     2     3     4
A    94     1    54    68
B    74    10    88    82
C    73    88     8    76
D    11    74    81    21;

BINARY VARIABLES Y(i,j);
VARIABLE Z;

EQUATIONS OBJ, C1, C2, C3, C4;

OBJ..   Z =E= sum((i,j), C(i,j)*Y(i,j));
C1(i).. sum(j, Y(i,j) ) =E= 1;
C2(j).. sum(i, Y(i,j) ) =E= 1;
C3(i).. sum(j$(C(i,j) gt 10), Y(i,j) ) =E= 1;
C4(j).. sum(i$(C(i,j) gt 10), Y(i,j) ) =E= 1;

OPTION LIMROW = 0;
OPTION LIMCOL = 0;
OPTION SOLPRINT = off;

MODEL P3A /OBJ, C1, C2/;
MODEL P3B /OBJ, C3, C4/;

SOLVE P3A USING MIP MINIMIZING Z;
DISPLAY Y.L;
SOLVE P3B USING MIP MINIMIZING Z;
DISPLAY Y.L;
```